
    README

Autonomous agents for Beam Bots, built on
the Jido agent framework.
Where bb_reactor answers "how do
I execute this workflow?", bb_jido lets you answer "what should I do next
to achieve this goal?" — your robot becomes a goal-directed agent that
observes the world via BB.PubSub and dispatches commands or workflows in
response.
Layered architecture
┌─────────────────────────────────────────────────┐
│  Jido Agent                                     │
│  - Observes via BB.PubSub (bridged to signals)  │
│  - Routes signals to actions                    │
│  - Emits further signals as directives          │
├─────────────────────────────────────────────────┤
│  bb_reactor workflows  +  BB commands           │
│  (executed via BB.Jido.Action.Reactor and       │
│   BB.Jido.Action.Command)                       │
└─────────────────────────────────────────────────┘
What's in the box
	Module	Purpose
	BB.Jido.Plugin.Robot	Jido v2 plugin you attach to an agent — adds robot state, the standard actions, default bb.* signal routes, and the PubSub bridge
	BB.Jido.Action.Command	Run a BB command (apply(robot, command, [goal])) and await its result
	BB.Jido.Action.Reactor	Run a bb_reactor workflow with the robot bound into context.private.bb_robot
	BB.Jido.Action.WaitForState	Wait for the robot state machine to enter a target state
	BB.Jido.Action.GetJointState	Read current joint positions/velocities
	BB.Jido.Action.SafetyAware	Mixin that aborts an action with {:safety_not_armed, state} unless the robot is armed
	BB.Jido.PubSubBridge	GenServer that forwards {:bb, path, %BB.Message{}} messages into an agent as Jido.Signals
	BB.Jido.Signal	Canonical BB.Message → Jido.Signal mapping (CloudEvents bb.* namespace)
	BB.Jido.Telemetry	Telemetry spans for actions and a per-signal counter

Installation
The fastest path is via Igniter — it adds the dep, generates a Jido
instance module, and wires it into your application supervision tree in
one command:
mix igniter.install bb_jido --robot MyApp.Robot

With --robot, it also scaffolds an agent module that attaches
BB.Jido.Plugin.Robot for that robot. Drop the flag to skip the agent
and add it later with mix bb_jido.add_agent.
bb_reactor is not a hard dependency. If you want to use
BB.Jido.Action.Reactor, add it the same way:
mix igniter.install bb_reactor

See the scaffolding how-to
for the full list of generators.
Quick start
After mix igniter.install bb_jido --robot MyApp.Robot, you'll have:
	lib/my_app/jido.ex — use Jido, otp_app: :my_app
	lib/my_app/robot/agent.ex — agent with the robot plugin attached
	lib/my_app/application.ex — children list gains {Jido, [name: MyApp.Jido]}

Start the agent and send it a signal:
{:ok, pid} = Jido.start_agent(MyApp.Jido, MyApp.Robot.Agent, id: "main")

:ok =
  Jido.AgentServer.cast(
    pid,
    Jido.Signal.new!(
      "bb.command.execute",
      %{robot: MyApp.Robot, command: :home, goal: %{}}
    )
  )
Signals
The PubSub bridge maps BB.PubSub events into Jido signals with a stable
naming scheme:
	Signal type	Source
	bb.state.transition	[:state_machine] topic, %BB.StateMachine.Transition{} payloads
	bb.safety.error	[:safety, :error] topic, %BB.Safety.HardwareError{} payloads
	bb.pubsub.<path>	Anything else — dotted source path (e.g. bb.pubsub.sensor.joint_state)

The source URI is /bb/<robot module> for traceability. Payload, path, and
robot are all available under signal.data.
By default the bridge only subscribes to [:state_machine]. Pass :topics
(and optionally :message_types or :throttle_ms) when attaching the
plugin to opt into higher-volume topics:
plugins: [
  {BB.Jido.Plugin.Robot,
   %{robot: MyRobot,
     topics: [[:state_machine], [:sensor, :joint_state]],
     throttle_ms: 100}}
]
Error taxonomy
Actions return a consistent tagged-error set:
	{:error, :safety_disarmed} — command exited because the robot was disarmed
	{:error, {:command_failed, reason}} — any other command failure
	{:error, {:reactor_failed, errors}} — reactor returned errors
	{:error, {:reactor_halted, reason}} — reactor was halted mid-flight
	{:error, {:safety_not_armed, state}} — BB.Jido.Action.SafetyAware
guard tripped

Documentation
Full documentation is available at HexDocs.


  

    Change Log

All notable changes to this project will be documented in this file.
See Conventional Commits for commit guidelines.
v0.1.2 (2026-05-20)
v0.1.1 (2026-05-19)
Features:
	initial implementation of proposal 0009 (bb_jido) (#11) by James Harton



  

    Your First Agent

In this tutorial you'll wire a Beam Bots robot up to a Jido agent and
ask the agent to execute a robot command by sending it a signal. By the
end you'll have a running supervision tree containing your robot, a Jido
instance, and one live agent.
Prerequisites
	Completed the Beam Bots tutorials, or at least
First Robot and
Commands and State Machine.
	An Elixir project that already declares a working robot module (we'll
call it MyRobot throughout).

For Roboticists: Where a behaviour tree would tick every frame, a
Jido agent stays idle until something happens. Signals are the inputs —
they're either casually sent by application code (like a teleop request)
or bridged in from BB's PubSub.

For Elixirists: A Jido agent is a GenServer plus a router. You give
it a list of plugins; each plugin contributes actions, signal routes, and
child processes. Sending a signal looks up an action in the router and
runs it inside the agent process.

Step 1: Install bb_jido
One command sets up everything you need:
mix igniter.install bb_jido --robot MyRobot

This:
	Adds {:bb_jido, "~> 0.1"} to your mix.exs.
	Creates lib/my_app/jido.ex — your application's Jido instance, the
supervisor under which agents will run.
	Adds {Jido, [name: MyApp.Jido]} to your application's supervision
tree.
	Creates lib/my_robot/agent.ex — an agent module that attaches
BB.Jido.Plugin.Robot for MyRobot.

After the installer runs you'll have:
# lib/my_app/jido.ex
defmodule MyApp.Jido do
  use Jido, otp_app: :my_app
end
# lib/my_robot/agent.ex
defmodule MyRobot.Agent do
  use Jido.Agent,
    name: "agent",
    plugins: [
      {BB.Jido.Plugin.Robot, %{robot: MyRobot}}
    ]
end
BB.Jido.Plugin.Robot is the integration point. Attaching it adds four
actions to the agent (BB.Jido.Action.Command, .Reactor, .WaitForState,
.GetJointState) and three default signal routes:
	Signal type	Action
	bb.command.execute	BB.Jido.Action.Command
	bb.reactor.run	BB.Jido.Action.Reactor
	bb.state.wait	BB.Jido.Action.WaitForState

The plugin also mounts a BB.Jido.PubSubBridge
under the agent so robot state-machine transitions arrive as signals.
We'll see that in Tutorial 2.
Want to do this by hand instead? The four generated pieces — the
dep, the Jido instance, the supervision tree entry, the agent module —
are all small. If you prefer to write them yourself, the equivalent
generators are mix bb_jido.add_jido_instance and
mix bb_jido.add_agent. See the
scaffolding how-to.

Step 2: Make sure the robot starts in simulation
Open lib/my_app/application.ex and pass simulation: :kinematic to your
robot's child spec so you can run without hardware while learning:
children = [
  {MyRobot, simulation: :kinematic},
  {Jido, [name: MyApp.Jido]}
]
(You can also use mix bb.add_robot --robot MyRobot if the robot module
doesn't yet exist.)
Step 3: Start an agent and send it a signal
Open iex -S mix. The robot and Jido instance start automatically. Spawn
an agent:
iex> {:ok, pid} = Jido.start_agent(MyApp.Jido, MyRobot.Agent, id: "main")
{:ok, #PID<0.500.0>}
The agent is now running. Robots start in :disarmed, so before we ask
the agent to do real work we'll arm the robot — using the agent:
iex> arm_signal =
...>   Jido.Signal.new!(
...>     "bb.command.execute",
...>     %{robot: MyRobot, command: :arm, goal: %{}}
...>   )
iex> Jido.AgentServer.cast(pid, arm_signal)
:ok
Jido.AgentServer.cast/2 enqueues the signal. The router matches
bb.command.execute to BB.Jido.Action.Command, which calls
MyRobot.arm(%{}) and awaits the result via BB.Command.await/2. After
a moment, the robot will be :armed and the action will have returned
{:ok, %{outcome: :armed, ...}}. You can verify:
iex> BB.Robot.Runtime.state(MyRobot)
:idle
Why :idle and not :armed? Robots have two independent states: the
safety state (:armed/:disarmed) and the operational state
(:idle/:executing). Runtime.state/1 returns the operational one when
safety is :armed. See Commands and State Machine.

Step 4: Run any robot command
Once armed, you can execute any command in the same way. If your robot
declares a :home command:
iex> Jido.AgentServer.cast(
...>   pid,
...>   Jido.Signal.new!(
...>     "bb.command.execute",
...>     %{robot: MyRobot, command: :home, goal: %{}}
...>   )
...> )
:ok
Step 5: Synchronous calls
cast/2 is fire-and-forget. If you want the result, use call/3 instead:
iex> {:ok, agent} =
...>   Jido.AgentServer.call(
...>     pid,
...>     Jido.Signal.new!(
...>       "bb.command.execute",
...>       %{robot: MyRobot, command: :home, goal: %{}}
...>     )
...>   )
The return is the updated agent struct; the action's {:ok, result} is
recorded in agent.result.
What you've built
Application
├── MyRobot                            # your robot + its supervision tree
└── Jido (name: MyApp.Jido)
    └── MyRobot.Agent (id: "main")
        ├── BB.Jido.PubSubBridge       # mounted by the plugin
        └── signal router → actions    # bb.command.execute, …
The agent is a single process; the bridge is a sibling under the agent's
own children. Crashes in the bridge restart the bridge only.
Where next
	Reacting to PubSub — read events into the
agent and react to them.
	Running Workflows — invoke a bb_reactor
workflow as a single Jido action.
	Layered architecture — why the agent
sits above bb_reactor rather than replacing it.
	Scaffolding with Igniter — all
four generator tasks.



  

    Reacting to PubSub

In Tutorial 1 you sent signals into an agent.
This tutorial shows the other direction: signals arriving from the robot.
Beam Bots publishes events on BB.PubSub (state-machine transitions,
sensor readings, safety errors); bb_jido bridges them into the agent as
Jido.Signals so the agent can route them to actions.
Prerequisites
	Tutorial 1.
	Familiarity with BB.PubSub (see Sensors and PubSub).

How the bridge works
When you attached BB.Jido.Plugin.Robot to your agent, the plugin mounted
a BB.Jido.PubSubBridge
under the agent's supervision tree. The bridge:
	Subscribes to BB topics on behalf of the agent (default:
[:state_machine]).
	Translates each {:bb, path, %BB.Message{}} delivery into a
Jido.Signal using BB.Jido.Signal.from_pubsub/3.
	Casts the signal into the agent via Jido.AgentServer.cast/2.

The signal's type follows the bb.* namespace:
	BB event	Signal type
	State machine transitions	bb.state.transition
	Safety hardware errors	bb.safety.error
	Anything else	bb.pubsub.<dotted.path>

Step 1: Watch transitions arrive
Add a debug subscription in iex so you can see what the bridge would
push at the agent:
iex> BB.PubSub.subscribe(MyRobot, [:state_machine])
{:ok, _}

iex> Jido.AgentServer.cast(
...>   pid,
...>   Jido.Signal.new!(
...>     "bb.command.execute",
...>     %{robot: MyRobot, command: :arm, goal: %{}}
...>   )
...> )
:ok

iex> flush()
{:bb, [:state_machine],
 %BB.Message{
   payload: %BB.StateMachine.Transition{from: :disarmed, to: :armed},
   ...
 }}
That same {:bb, ...} tuple is what BB.Jido.PubSubBridge receives.
Step 2: Route the transition to your own action
Scaffold the action with the generator:
mix bb_jido.add_action MyRobot.OnArmed \\
  --description "Runs once the robot finishes arming"

That creates lib/my_robot/on_armed.ex with a stub run/2. Replace the
stub so it inspects the transition payload — Jido v2 passes signal.data
into the action as its params, so the schema declares fields that match
the shape BB.Jido.Signal produces:
defmodule MyRobot.OnArmed do
  use Jido.Action,
    name: "on_armed",
    description: "Runs once the robot finishes arming",
    schema: [
      robot: [type: :atom, required: true],
      path: [type: {:list, :atom}, required: true],
      message: [type: :any, required: true]
    ]

  @impl Jido.Action
  def run(%{robot: robot, message: %BB.Message{payload: transition}}, _context) do
    IO.puts("#{inspect(robot)} transitioned #{transition.from} → #{transition.to}")
    {:ok, %{to: transition.to}}
  end
end
For Elixirists: A Jido action is a behaviour module with a run/2
callback. The schema: declares which params it accepts. When the
router dispatches a signal to an action, signal.data is what gets
passed as params — so the schema fields match the keys in data.

Now expose the action via a tiny plugin that adds the route. Plugins are
small enough that we don't ship a generator for them — write it by hand:
defmodule MyRobot.WatcherPlugin do
  use Jido.Plugin,
    name: "watcher",
    state_key: :watcher,
    actions: [MyRobot.OnArmed],
    signal_routes: [
      {"bb.state.transition", MyRobot.OnArmed}
    ]
end
And attach it alongside the robot plugin. Edit the generated
lib/my_robot/agent.ex to add the watcher:
defmodule MyRobot.Agent do
  use Jido.Agent,
    name: "agent",
    plugins: [
      {BB.Jido.Plugin.Robot, %{robot: MyRobot}},
      MyRobot.WatcherPlugin
    ]
end
Restart your iex session and arm the robot again — the bridge forwards the
transition, the router dispatches to MyRobot.OnArmed, and you'll see:
Robot transitioned disarmed → armed
Step 3: Filter by which transition
Two bb.state.transition signals arrive when you cycle arm → disarm.
If you only care about one, pattern-match in your action:
@impl Jido.Action
def run(%{message: %BB.Message{payload: payload}} = params, _context) do
  case payload do
    %BB.StateMachine.Transition{to: :armed} ->
      handle_armed(params.robot)

    %BB.StateMachine.Transition{to: :error} ->
      handle_error(params.robot)

    _other ->
      {:ok, %{ignored: true}}
  end
end
Why not route on a more specific type? Today the bridge maps every
transition to the same bb.state.transition type. Discriminating
further lives in your action — that keeps the signal vocabulary stable.

Step 4: Subscribe to additional topics
By default the plugin only subscribes to [:state_machine]. To also
forward, say, joint state messages, pass :topics when attaching:
plugins: [
  {BB.Jido.Plugin.Robot,
   %{
     robot: MyRobot,
     topics: [[:state_machine], [:sensor, :joint_state]],
     message_types: [BB.StateMachine.Transition, BB.Sensor.JointState]
   }}
]
Now joint-state messages will arrive as bb.pubsub.sensor.joint_state
signals. Be careful: BB sensors can publish at 100Hz. The next tutorial
and the Throttle high-volume signals
guide show how to keep the agent mailbox sane.
What you've built
BB.PubSub                       Jido agent
─────────                       ──────────
state_machine ──▶ Bridge ──▶ bb.state.transition ──▶ MyRobot.OnArmed
                            ▶ (any other route)
The bridge is just plumbing. All decisions about what to do with an event
live in your actions and plugins.
Where next
	Running Workflows — invoke a reactor when a
signal arrives.
	Signals and PubSub — the design
rationale for the bridge.
	Throttle high-volume signals
— keep the mailbox quiet under 100Hz sensors.



  

    Running Workflows

BB.Jido.Action.Command runs a single robot command. For more involved
sequences — pick-and-place, calibration, an assembly step — you almost
always want a bb_reactor workflow. This tutorial shows how to invoke a
reactor from an agent: the agent decides "this is what I want to do
next", the reactor handles the structured execution and compensation.
Prerequisites
	Tutorials 1 and 2.

	Familiarity with bb_reactor.

	Install it with Igniter (bb_jido doesn't depend on it):
mix igniter.install bb_reactor



Step 1: Define a reactor
Workflows are reactor modules. A minimal one that runs a single command:
defmodule MyRobot.Workflow.PickAndPlace do
  use Reactor

  middlewares do
    middleware BB.Reactor.Middleware.Context
  end

  input :pick_pose
  input :place_pose

  step :move_to_pick do
    impl {BB.Reactor.Step.Command, command: :move_to}
    argument :target, input(:pick_pose)
  end

  step :grasp do
    impl {BB.Reactor.Step.Command, command: :close_gripper}
    wait_for :move_to_pick
  end

  step :move_to_place do
    impl {BB.Reactor.Step.Command, command: :move_to}
    argument :target, input(:place_pose)
    wait_for :grasp
  end

  step :release do
    impl {BB.Reactor.Step.Command, command: :open_gripper}
    wait_for :move_to_place
  end

  return :release
end
The BB.Reactor.Middleware.Context middleware reads
context.private.bb_robot and exposes it to every step.
Step 2: Invoke the reactor from the agent
BB.Jido.Action.Command's reactor cousin is BB.Jido.Action.Reactor. It's
already on the agent — the robot plugin attaches it. Send it via the
default route, bb.reactor.run:
:ok =
  Jido.AgentServer.cast(
    pid,
    Jido.Signal.new!(
      "bb.reactor.run",
      %{
        robot: MyRobot,
        reactor: MyRobot.Workflow.PickAndPlace,
        inputs: %{
          pick_pose: %{x: 0.2, y: 0.0, z: 0.1},
          place_pose: %{x: 0.0, y: 0.2, z: 0.1}
        }
      }
    )
  )
That's it. The action injects context.private.bb_robot = MyRobot and
calls Reactor.run/3. On success it returns
{:ok, %{robot: ..., reactor: ..., result: result}} to the agent.
Why didn't I have to thread the robot module through every step?
Because the reactor middleware reads it from context. Each step picks
the robot out of context.private.bb (a %BB.Reactor.Context{}).

Step 3: Compose decisions before invoking
The whole point of agents is to decide before running a structured
workflow. Scaffold a higher-level action that wraps the reactor call:
mix bb_jido.add_action MyRobot.Actions.PickRedBlock

Then replace the stub run/2 so it selects inputs and invokes the
reactor:
defmodule MyRobot.Actions.PickRedBlock do
  use Jido.Action,
    name: "pick_red_block",
    schema: [robot: [type: :atom, required: true]]

  alias Jido.Agent.Directive.Emit

  @impl Jido.Action
  def run(%{robot: robot}, _context) do
    pick_pose = locate_red_block(robot)
    place_pose = %{x: 0.0, y: 0.2, z: 0.1}

    BB.Jido.Action.Reactor.run(
      %{
        robot: robot,
        reactor: MyRobot.Workflow.PickAndPlace,
        inputs: %{pick_pose: pick_pose, place_pose: place_pose}
      },
      %{}
    )
  end

  defp locate_red_block(_robot), do: %{x: 0.2, y: 0.0, z: 0.1}
end
You can call BB.Jido.Action.Reactor.run/2 directly — actions are plain
Elixir modules.
Step 4: Handle reactor outcomes
BB.Jido.Action.Reactor maps reactor's three return shapes onto
bb_jido's error taxonomy:
	Reactor returns	Action returns
	{:ok, result}	{:ok, %{result: result, ...}}
	{:ok, result, _struct}	{:ok, %{result: result, ...}}
	{:halted, halted}	{:error, {:reactor_halted, halted}}
	{:error, errors}	{:error, {:reactor_failed, errors}}

If your workflow has compensation steps, they run before
{:error, ...} is returned — that's the reactor's saga behaviour, not
the action's job.
Should the agent retry or compensate? That's an application
decision. The reactor unwinds its own steps; the agent decides what
happens next (re-plan, escalate, ask a human). See Layered
architecture.

Step 5: Chain the result into another signal
Actions can emit signals via directives. To follow a successful pick with
a celebratory state-machine transition, return an Emit directive:
alias Jido.Agent.Directive.Emit

def run(params, _ctx) do
  case BB.Jido.Action.Reactor.run(params, %{}) do
    {:ok, result} ->
      followup =
        Jido.Signal.new!("my_robot.pick.completed", %{
          target: params[:target]
        })

      {:ok, result, %Emit{signal: followup}}

    {:error, reason} ->
      {:error, reason}
  end
end
The runtime dispatches the follow-up signal back through the router. Any
plugin that listens for my_robot.pick.completed will fire.
What you've built
bb.reactor.run signal
        │
        ▼
BB.Jido.Action.Reactor
        │  Reactor.run/3 with private.bb_robot = MyRobot
        ▼
PickAndPlace ── steps run BB commands via BB.Reactor.Step.Command
        │
        ▼
{:ok, result}  →  agent
        │
        ▼ (optional directive)
my_robot.pick.completed signal  →  back into the router
Where next
	Layered architecture — why the
agent dispatches reactors rather than running step logic directly.
	Emit directives from actions
— more on signal chaining.
	Wait for robot state — block
inside an action until the robot reaches a given state.



  

    How to bridge additional PubSub topics

By default BB.Jido.Plugin.Robot only subscribes its PubSubBridge to
[:state_machine]. This guide shows how to forward additional BB topics
into your agent.
When to do this
Add a topic when an action or plugin in the agent needs to react to
something other than state-machine transitions — for example a sensor
reading, a safety error, or a parameter change.
Pass :topics when attaching the plugin
defmodule MyRobot.Agent do
  use Jido.Agent,
    name: "my_robot",
    plugins: [
      {BB.Jido.Plugin.Robot,
       %{
         robot: MyRobot,
         topics: [
           [:state_machine],
           [:safety, :error],
           [:sensor, :force_torque]
         ]
       }}
    ]
end
Each entry is a list of atoms — the same format that
BB.PubSub.subscribe/3 accepts. The list is appended on top of the BB
defaults, not merged, so include [:state_machine] explicitly if you
still want it.
Restrict by payload type
Subscribing to a path delivers every payload published on that path or
its descendants. Restrict at the PubSub layer with :message_types:
{BB.Jido.Plugin.Robot,
 %{
   robot: MyRobot,
   topics: [[:sensor]],
   message_types: [BB.Sensor.JointState, BB.Sensor.ForceTorque]
 }}
The filter is per-bridge (one bridge per agent), so it applies to every
subscribed topic.
Check the resulting signal types
Each forwarded message becomes a Jido.Signal with a stable type string:
	Payload module	Signal type
	BB.StateMachine.Transition	bb.state.transition
	BB.Safety.HardwareError	bb.safety.error
	Anything else	bb.pubsub.<dotted source path>

So a sensor publishing on [:sensor, :force_torque] becomes
bb.pubsub.sensor.force_torque. Subscribe an action to it via:
signal_routes: [
  {"bb.pubsub.sensor.force_torque", MyRobot.Actions.HandleFT}
]
The signal carries the original %BB.Message{} in signal.data.message.
See also
	Reacting to PubSub — end-to-end
walkthrough.
	Throttle high-volume signals — keep
100Hz topics from drowning the agent.
	Signal types reference — full type table.



  

    How to emit signals from an action

A Jido action can return directives alongside its result. The most
useful one for bb_jido is Jido.Agent.Directive.Emit, which dispatches
a follow-up signal. This lets an action say "I finished — now do this
next" without imperatively calling other actions.
The basic shape
Scaffold the action with mix bb_jido.add_action MyRobot.Actions.PickThenAnnounce,
then replace the stub run/2:
defmodule MyRobot.Actions.PickThenAnnounce do
  use Jido.Action,
    name: "pick_then_announce",
    schema: [robot: [type: :atom, required: true], target: [type: :atom, required: true]]

  alias Jido.Agent.Directive.Emit

  @impl Jido.Action
  def run(%{robot: robot, target: target}, _context) do
    case BB.Jido.Action.Reactor.run(
           %{
             robot: robot,
             reactor: MyRobot.Workflow.Pick,
             inputs: %{target: target}
           },
           %{}
         ) do
      {:ok, result} ->
        followup =
          Jido.Signal.new!("my_robot.pick.completed", %{target: target})

        {:ok, result, %Emit{signal: followup}}

      {:error, reason} ->
        failure =
          Jido.Signal.new!("my_robot.pick.failed", %{target: target, reason: reason})

        {:ok, %{target: target, recovered: false}, %Emit{signal: failure}}
    end
  end
end
The Emit directive is the third element of the return tuple. The
runtime dispatches signal back through the agent's router, where any
matching signal_routes will fire.
Multiple directives
Return a list to emit several signals:
{:ok, result,
 [
   %Emit{signal: log_signal},
   %Emit{signal: followup_signal}
 ]}
Pick a dispatch target
By default the runtime dispatches to the agent itself (self()). Pass
:dispatch on the Emit struct to send elsewhere:
%Emit{
  signal: signal,
  dispatch: {:pubsub, topic: "robot_events"}
}
See Jido.Signal.Dispatch for the list of adapters (:pid, :pubsub,
:bus, :http, :logger, etc.).
Don't bypass the router for cross-action chaining
You could call MyRobot.Actions.Next.run(params, %{}) directly from
inside your action — actions are plain modules. Prefer emitting a signal
when:
	The next action lives in a different plugin.
	You want telemetry/observability on the hand-off.
	Another listener might also want to react (multi-cast).

Direct calls are fine when the chain is private to one module and you
don't want it observable.
See also
	Jido.Agent.Directive
— the full directive vocabulary (Emit, Spawn, Schedule, Stop, …).
	Running Workflows — the
end-to-end example uses Emit to chain reactor success into a signal.



  

    How to scaffold bb_jido modules with Igniter

bb_jido ships four Igniter tasks for the boilerplate parts of wiring
agents up: declaring a Jido instance, adding it to the supervision tree,
creating an agent, and creating individual actions.
One-shot install
mix igniter.install bb_jido runs bb_jido.install, which always
declares a Jido instance and (when --robot is given) scaffolds an
agent for it:
mix igniter.install bb_jido --robot MyApp.Robot

After that:
lib/my_app/jido.ex          ← use Jido, otp_app: :my_app
lib/my_app/robot/agent.ex   ← use Jido.Agent with BB.Jido.Plugin.Robot
lib/my_app/application.ex   ← children list gains {Jido, [name: MyApp.Jido]}
You're now one Jido.start_agent/3 call away from a running agent.
Individual tasks
mix bb_jido.add_jido_instance
Creates the Jido instance module and adds {Jido, name: <module>} to the
application's supervision tree.
mix bb_jido.add_jido_instance
mix bb_jido.add_jido_instance --jido-instance MyApp.AgentRuntime

Idempotent — running it again is a no-op.
mix bb_jido.add_agent
Creates an agent module that attaches BB.Jido.Plugin.Robot for the given
robot.
mix bb_jido.add_agent --robot MyApp.Robot
mix bb_jido.add_agent --robot MyApp.Robot --agent MyApp.MainAgent --name main_robot

	Flag	Default	Notes
	--robot	{AppPrefix}.Robot	Robot module the agent drives
	--agent	{robot}.Agent	Module name for the agent
	--name	snake_cased last segment	Jido name: string

The task does not start the agent — that's a runtime call:
Jido.start_agent(MyApp.Jido, MyApp.Robot.Agent, id: "main")
mix bb_jido.add_action
Scaffolds a Jido action with a run/2 stub returning {:ok, %{}}.
mix bb_jido.add_action MyApp.Actions.Pick
mix bb_jido.add_action MyApp.Actions.MovePose --safety-aware
mix bb_jido.add_action MyApp.Actions.Teleop --name teleop_step \\
  --description "Drive the robot from a teleop joystick"

--safety-aware mixes in BB.Jido.Action.SafetyAware
and seeds the schema with a :robot field, so the action refuses to run
unless BB.Safety.state(robot) == :armed.
	Flag	Default	Notes
	--name	snake_cased last segment	Jido name: string
	--description	(none)	Jido description: string
	--safety-aware	false	Add the safety guard mixin

What the tasks don't do
	They don't add bb_reactor as a dependency — bb_jido doesn't depend
on it, and not every agent needs workflows.
	They don't patch existing plugins: lists or signal_routes: maps —
once an agent exists, hand-editing is straightforward and avoids the
fragility of source patching.
	They don't generate example apps or a runnable simulation — the
inline tutorials cover that path.



  

    How to throttle high-volume signals

BB sensors can publish at 100Hz or higher. Forwarding every reading into
the agent will overflow its mailbox and starve other work. This guide
shows the layered mitigations available in bb_jido.
Layer 1: don't subscribe in the first place
The cheapest filter is the one that never reaches the bridge. Only list
topics in the plugin config that the agent actually needs:
{BB.Jido.Plugin.Robot,
 %{
   robot: MyRobot,
   topics: [[:state_machine]]   # joint state etc. NOT subscribed
 }}
Layer 2: filter by message type at subscribe time
If you need a path but only certain payloads on it, pass :message_types.
The filter runs inside BB.PubSub before the message is sent at all:
{BB.Jido.Plugin.Robot,
 %{
   robot: MyRobot,
   topics: [[:sensor]],
   message_types: [BB.Sensor.ForceTorque]
 }}
Now JointState publishes on [:sensor, :joint_state] won't reach the
bridge at all even though [:sensor] is subscribed.
Layer 3: throttle in the bridge
For genuinely high-volume topics, set :throttle_ms. The bridge will drop
any signal of a type that was emitted less than throttle_ms milliseconds
ago:
{BB.Jido.Plugin.Robot,
 %{
   robot: MyRobot,
   topics: [[:state_machine], [:sensor, :joint_state]],
   throttle_ms: 100
 }}
A 100Hz joint-state stream becomes a 10Hz signal stream. State-machine
transitions still pass through immediately because they have a different
signal type (bb.state.transition vs bb.pubsub.sensor.joint_state).
Important: the throttle drops by type, not by content. A latest-
value workflow is fine; a "we must see every reading" workflow is not.
For that, subscribe directly to BB.PubSub outside the agent.

Layer 4: don't put fast events in the agent at all
If an event is high-volume and doesn't affect agent decisions, keep it
out of the agent. The pattern:
	A separate GenServer subscribes to the high-volume topic.
	It maintains the current value and exposes it via :ets or a public
API.
	The agent reads the value when it needs to (via an action that does a
cheap lookup).

This is exactly the "last-known-state cache" the proposal recommends.
BB.Jido.Action.GetJointState is an instance of this pattern: it reads
positions from BB.Robot.Runtime's ETS cache rather than waiting for a
PubSub message.
See also
	BB.Jido.PubSubBridge
— full option reference.
	Signals and PubSub — why throttling
on the bridge side is preferable to backpressure at the mailbox.



  

    How to wait for a robot state

Sometimes you want an agent action to block until the robot enters a
particular state — for example, "wait until the robot is :idle before
issuing the next move". This guide covers the two available patterns.
Option 1: BB.Jido.Action.WaitForState
The simplest case is a single, foreground wait. The action is on the agent
already (the robot plugin attaches it). Send it via the default route:
:ok =
  Jido.AgentServer.cast(
    pid,
    Jido.Signal.new!(
      "bb.state.wait",
      %{robot: MyRobot, target: :idle, timeout: 5_000}
    )
  )
Return shapes:
	Outcome	Return
	Robot already in target	{:ok, %{robot: ..., state: target}}
	Transitions into target within timeout	{:ok, %{robot: ..., state: target}}
	timeout elapses first	{:error, :timeout}

The action subscribes to [:state_machine] for the duration of the wait
and unsubscribes when it returns.
Caveat: this blocks the agent process while it waits. If your
agent also needs to react to other signals during the wait, that
processing is paused. For non-blocking waits, see Option 2.

Option 2: an event-driven decision
For long waits or concurrent waits, react to the transition signal
instead. The robot plugin already forwards bb.state.transition signals
into the agent; scaffold an action that pattern-matches on the payload:
mix bb_jido.add_action MyRobot.OnTransition

defmodule MyRobot.OnTransition do
  use Jido.Action,
    name: "on_transition",
    schema: [
      robot: [type: :atom, required: true],
      path: [type: {:list, :atom}, required: true],
      message: [type: :any, required: true]
    ]

  @impl Jido.Action
  def run(%{robot: robot, message: %BB.Message{payload: payload}}, _context) do
    case payload do
      %BB.StateMachine.Transition{to: :idle} -> handle_idle(robot)
      %BB.StateMachine.Transition{to: :error} -> handle_error(robot)
      _ -> {:ok, %{ignored: true}}
    end
  end
end
Attach via a plugin with signal_routes: [{"bb.state.transition", MyRobot.OnTransition}].
The agent stays responsive throughout — the action runs only when the
transition actually happens, and only for the transitions you care about.
Rule of thumb: if the timeout is short (sub-second), Option 1 is
fine. If it's seconds or more, prefer Option 2.

Pre-cached state on the plugin
BB.Jido.Plugin.Robot keeps a safety_state field in its plugin state
that's updated whenever a transition signal arrives. Read it from another
action via context.agent.state.robot.safety_state to skip a Runtime
lookup. This is a convenience cache — BB.Robot.Runtime.state/1 is also
cheap (ETS), so use whichever reads better in your code.
See also
	BB.Jido.Action.WaitForState
	Reacting to PubSub



  

    How to write a safety-aware action

A safety-aware action refuses to run unless BB.Safety.state(robot) == :armed.
This guide shows the recommended pattern: use BB.Jido.Action.SafetyAware.
When to do this
Use the safety guard for any action that physically moves the robot.
For read-only actions (BB.Jido.Action.GetJointState, perception, logging)
the guard is unnecessary overhead.
The recipe
Scaffold the action with --safety-aware:
mix bb_jido.add_action MyRobot.Actions.MoveSomewhere \\
  --safety-aware \\
  --description "Drive the robot to a configured pose"

The generator produces a module with the mixin already in place. Add
your own fields to the schema and fill in run/2:
defmodule MyRobot.Actions.MoveSomewhere do
  use Jido.Action,
    name: "move_somewhere",
    description: "Drive the robot to a configured pose",
    schema: [
      robot: [type: :atom, required: true],
      pose: [type: :map, required: true]
    ]

  use BB.Jido.Action.SafetyAware

  @impl Jido.Action
  def run(%{robot: robot, pose: pose}, _context) do
    BB.Jido.Action.Command.run(
      %{robot: robot, command: :move_to, goal: pose},
      %{}
    )
  end
end
SafetyAware wraps your run/2 at compile time. The guard is invoked
before yours, so by the time MyRobot.Actions.MoveSomewhere.run/2 runs
the robot is guaranteed to be armed.
Error shapes
	Robot state	Return
	:armed	passes through to your run/2
	:disarmed / :disarming / :error	{:error, {:safety_not_armed, state}}
	robot not in params or context	{:error, :robot_not_specified}

Where the robot module is looked up
SafetyAware checks two places, in order:
	params[:robot] — the value passed via the action's schema.
	context[:robot] — useful when a parent action injects context.

If your action doesn't accept a :robot field in its schema, set it from
elsewhere into context before invoking, e.g.:
MyRobot.Actions.MoveSomewhere.run(%{pose: pose}, %{robot: MyRobot})
Don't double-guard
BB.Jido.Action.Command already maps :disarmed exits onto
{:error, :safety_disarmed}. The safety guard adds an early refusal so
the command is never even started — useful if starting the command would
itself have side effects (logging, telemetry, allocating resources).
See also
	BB.Jido.Action.SafetyAware
— error reference.
	Understanding safety —
the BB-level safety model.



  

    The Layered Architecture

bb_jido doesn't replace either bb or bb_reactor. It sits above
them. Understanding the three layers — and what belongs in which —
prevents most of the structural confusion teams encounter when first
introducing agents.
The three layers
┌─────────────────────────────────────────────────┐
│  Jido Agent                                     │  ← decide what to do
│  - Receives signals (PubSub events, user input) │
│  - Routes them to actions                       │
│  - Decides which workflow or command to invoke  │
├─────────────────────────────────────────────────┤
│  bb_reactor Workflows                           │  ← how to do it
│  PickAndPlace, Calibrate, ReturnHome            │
│  - Declared sequences with compensation         │
│  - Compile-time validated                       │
├─────────────────────────────────────────────────┤
│  BB Commands                                    │  ← do it
│  move_to_pose, close_gripper, home, arm         │
│  - Direct hardware control                      │
└─────────────────────────────────────────────────┘
Each layer answers a different question:
	Layer	Question
	Agent	"What should I do next to achieve this goal?"
	Reactor	"How do I execute this multi-step task safely?"
	Command	"How do I make the robot do this one thing?"

When to add an agent
You don't need an agent for every robot. Many production robots are happy
with bb plus bb_reactor — a reactor is run, it executes, returns a
result. Done.
Add an agent when the decision about which workflow to run is itself
non-trivial:
	The next action depends on perception ("if the part is here, pick it;
if it's missing, search").
	The next action depends on coordination ("if robot A has the part,
fetch from station B").
	The next action depends on input ("the operator just said 'home'").
	You need adaptive recovery ("the reactor failed — what now?").

Agents don't make sense if your decision logic is "always run reactor X
on a button press". A simple MyRobot.Workflows.run/2 function is fine.
Why the agent doesn't run steps directly
A Jido agent could dispatch BB commands one at a time, threading state
through with Emit directives. People sometimes try this because it
"looks like" a behaviour tree. Don't.
	Compensation is hard to express. Reactor's saga semantics roll back
partial work on failure. An agent emitting individual commands has to
reinvent that for every flow.
	Validation is hard. Reactor catches type and dependency mistakes at
compile time. Agents are runtime-only.
	The agent process becomes a hot path. Every step blocks the agent
mailbox. Even at modest rates this starves signal processing.

Use the agent to pick which reactor to run. Use the reactor to run it.
Where state lives
	State	Owner	Where
	Joint positions	BB runtime (ETS)	BB.Robot.Runtime.positions/1
	Safety state	BB safety controller (ETS)	BB.Safety.state/1
	Last transition (cached)	Robot plugin in agent state	agent.state.robot.safety_state
	Reactor intermediate results	Reactor (ephemeral)	context per step
	Application goals/queues	Your plugins	plugin state slice

The agent is a poor place to store anything BB already tracks — ETS reads
are cheap and authoritative. The agent's plugin state is the right home
for things that are agent-level: a pending-task queue, an active goal,
the last command's outcome, etc.
Multi-robot coordination
One agent per robot is the recommended default — supervision mirrors
robot identity, state stays local. Coordinator agents exist alongside,
not instead. They subscribe to robot-level signals (typically via
PubSub or your own bus) and emit task-assignment signals back.
Robot A agent ─▶ "robot.task.completed" ─▶ Coordinator ─▶ "robot.task.assigned" ─▶ Robot B agent
Keep the coordinator's per-robot state out of the per-robot plugins.
Why a separate package?
The proposal calls this out: not every robot needs agents, Jido adds
weight, and agent patterns are still evolving. Keeping the agent layer in
its own package means you can upgrade or replace it without touching the
rest of the stack.
See also
	Signals and PubSub — how events cross between
layers.
	Plugin lifecycle — when mount/2 and
child_spec/1 fire.
	The proposal — 0009-bb-jido.md in the proposals repo.



  

    Plugin Lifecycle

This page traces what happens when you attach BB.Jido.Plugin.Robot to
an agent and start the agent. Understanding the order lets you reason
about where to put state, why child_spec/1 can call self(), and how
the bridge stays scoped to the agent.
Where the agent lives
Application supervisor
└── Jido (name: MyApp.Jido)            ← DynamicSupervisor
    └── Jido.AgentServer (id: "main")  ← started by Jido.start_agent/3
        └── BB.Jido.PubSubBridge       ← started by plugin child_spec/1
The Jido instance is a DynamicSupervisor. Jido.start_agent/3 calls
DynamicSupervisor.start_child/2, which spawns a Jido.AgentServer
process. That AgentServer then runs the plugin lifecycle below.
Phase 1: agent compile-time
When you write:
use Jido.Agent,
  name: "my_robot",
  plugins: [{BB.Jido.Plugin.Robot, %{robot: MyRobot}}]
…Jido validates the plugin list at compile time and stores plugin specs
on the agent module. The robot module reference is captured in the spec
as data; nothing is started yet.
Phase 2: Jido.start_agent/3
The AgentServer is spawned. Its init/1 does two things relevant to
plugins:
	Calls each plugin's mount/2 to build the agent's initial state. This
is pure — no processes, no side effects.
	Schedules handle_continue(:post_init, state) for child startup.

BB.Jido.Plugin.Robot.mount/2:
def mount(_agent, %{robot: robot}) do
  {:ok,
   %{robot: robot, safety_state: :unknown, last_joint_state: %{}}}
end
The map returned becomes agent.state.robot (because the plugin's
state_key: :robot). If :robot is missing from config, mount/2
returns {:error, ...} and the agent fails to start.
Phase 3: handle_continue(:post_init, ...)
This is where children are started. Crucially, the work happens inside
the AgentServer process — so self() here is the agent's pid.
Jido.AgentServer.start_plugin_children/1 walks the plugin specs and
calls each plugin's child_spec/1. For BB.Jido.Plugin.Robot:
def child_spec(config) do
  agent_pid = self()        # ← captured at this moment
  # ...build PubSubBridge child spec with agent: agent_pid
end
This is the point: child_spec/1 is called from the agent process, so
self() is the agent. We capture it once and pass it to the bridge's
start_link/1 opts.
The returned spec is fed into Supervisor.child_spec(...)-style startup.
The bridge is now a monitored child of the AgentServer: if either
crashes, the supervision tree handles it.
Phase 4: bridge init/1
The bridge subscribes to its configured topics:
for topic <- topics do
  BB.PubSub.subscribe(robot, topic, message_types: message_types)
end
…and stashes the agent pid in its state. From now on:
BB.PubSub  ──[:bb, path, %BB.Message{}]──▶  Bridge
                                              │
                                              │ Jido.AgentServer.cast/2
                                              ▼
                                          AgentServer
The bridge sees every matching delivery, turns it into a signal, and
casts. The agent's router takes it from there.
Phase 5: signal routing
When the bridge casts a signal to the agent:
	Each plugin's handle_signal/2 pre-routing hook fires in declaration
order. BB.Jido.Plugin.Robot.handle_signal/2 watches for
bb.state.transition and updates its cached safety_state — that's
why the cache stays current without a separate subscriber.
	The signal router matches the type against the plugin's
signal_routes: and any other plugin's routes.
	The matched action's run/2 is invoked.
	Any returned directives (e.g. Emit) are dispatched.

What this means in practice
	Don't start processes in mount/2. It's pure; failures there crash
agent creation, not a restartable child.
	Don't store the agent pid in plugin state. It's not needed —
actions get the agent via context. The bridge needs it only because
it lives in a separate process.
	Don't restart the agent to pick up new bridge config. Stop and
restart the agent (Jido.stop_agent/2 then Jido.start_agent/3).
The bridge restarts as part of that.

See also
	BB.Jido.Plugin.Robot — config
reference.
	Layered architecture — where the agent sits
relative to BB and bb_reactor.
	Jido's plugin documentation
— the full callback list (we use a small subset).



  

    Signals and PubSub

BB and Jido both have an event mechanism. BB has BB.PubSub — a
hierarchical-path message broker. Jido has signals — CloudEvents-flavoured
structs routed by an agent's signal router. bb_jido translates the
former into the latter. This page explains why, and what the design
choices buy you.
Two systems, two vocabularies
BB.PubSub delivers Erlang messages with a uniform shape:
{:bb, source_path, %BB.Message{payload: payload, robot: robot, ...}}
	The source path is a list of atoms — the publisher's full path
(e.g. [:sensor, :joint_state]).
	The payload is a domain struct: BB.StateMachine.Transition,
BB.Safety.HardwareError, sensor-specific structs, etc.

Jido signals are CloudEvents: a :type string plus a payload in :data,
plus standard envelope fields (:source, :id, :specversion, :time).
The agent's router dispatches by type-string pattern.
These two vocabularies don't translate one-to-one. PubSub is a transport
— it doesn't care what the payload means. Signals are a protocol — the
type carries semantic meaning.
Why a bridge instead of a Jido Sensor?
Jido provides a Jido.Sensor abstraction for ingesting external events.
On paper that's the obvious place to hook PubSub in. In practice it adds
a layer of indirection that buys nothing:
	The sensor runtime expects events to be injected via
Jido.Sensor.Runtime.event/2. But BB already delivers events as Erlang
messages.
	We'd be writing a sensor that does nothing but receive a message and
forward it. The bridge code is the same; the sensor wrapper is dead
weight.

So BB.Jido.PubSubBridge is a plain GenServer that subscribes to
BB.PubSub and casts forward via Jido.AgentServer.cast/2. A future
sensor wrapper remains possible if a future Jido feature genuinely needs
one — it's not a one-way door.
Canonical signal types
The bridge produces three families of signal type:
	When	Type	Why
	%BB.StateMachine.Transition{} payload	bb.state.transition	Specialised — agents almost always want to react to transitions specifically.
	%BB.Safety.HardwareError{} payload	bb.safety.error	Specialised — safety errors deserve their own type.
	Anything else	bb.pubsub.<dotted source path>	Generic — preserves the path information but no semantic claim.

The specialised types are stable. Even if a future BB version changes
where state transitions are published, the signal type stays
bb.state.transition. The generic bb.pubsub.* type is necessarily
coupled to the path layout — it's a fallback, not a contract.
The signal :source is /bb/<robot module>, following CloudEvents'
URI-like source convention. Traceability stays sane when multiple robots'
events end up on the same downstream bus.
Filtering happens before the agent
Three places to filter, from cheapest to most expensive:
	Topic allowlist — only subscribe to paths the agent needs. Filters
inside BB's registry; non-matching topics never trigger any work.
	Message-type filter — pass :message_types so the registry only
delivers matching payloads.
	Bridge-side throttle — :throttle_ms drops repeat signals of the
same type within the window. Signals are still constructed; they're
just not cast forward.

Filtering at the agent's mailbox is the wrong place. By the time the
agent receives the cast, it's already paid the signal-construction cost,
and the mailbox queue may already be backed up. The bridge owns the
discipline.
Filtering happens outside PubSub for content
The bridge can't peek inside a payload — that would couple it to every
payload type in BB. If you need content-based filtering ("only IMU
readings on link 3 with temperature > 60"), do it in your action:
def run(%{message: %BB.Message{payload: payload}} = params, _ctx) do
  case payload do
    %BB.Sensor.IMU{frame_id: :link3, temperature: t} when t > 60 ->
      handle_overheat(params.robot)

    _ ->
      {:ok, %{ignored: true}}
  end
end
This keeps the bridge a thin pipe and pushes domain knowledge into the
agent's action code, where it belongs.
See also
	Bridge additional PubSub topics
— the operational steps.
	Signal types reference — full table of
what gets emitted.
	BB's PubSub documentation.



  

    Error Taxonomy

bb_jido actions return errors as {:error, reason} tuples. The
reason follows a small, stable set of tags so application code can
pattern-match without inspecting strings.
Action errors
	Error	When	Returned by
	:safety_disarmed	Command process exited with :disarmed (robot was disarmed mid-flight).	BB.Jido.Action.Command
	{:command_failed, reason}	Any other command failure: command callback returned an error, exited unexpectedly, timed out, or apply(robot, command, [goal]) itself failed.	BB.Jido.Action.Command
	{:reactor_failed, errors}	Reactor returned {:error, errors} — at least one step failed and any defined compensation has already run.	BB.Jido.Action.Reactor
	{:reactor_halted, halted}	Reactor returned {:halted, halted} — execution was paused (e.g. a step asked to halt). halted is the halted reactor struct.	BB.Jido.Action.Reactor
	:timeout	WaitForState ran out of time before the target state was reached.	BB.Jido.Action.WaitForState
	{:subscribe_failed, reason}	BB.PubSub.subscribe/3 refused the subscription.	BB.Jido.Action.WaitForState

BB.Jido.Action.SafetyAware guard errors
	Error	Meaning
	{:safety_not_armed, state}	The robot's safety state is state (one of :disarmed, :disarming, :error). The wrapped run/2 was not invoked.
	:robot_not_specified	Neither params[:robot] nor context[:robot] provided a robot module.

Pattern-matching cheatsheet
case BB.Jido.Action.Command.run(params, %{}) do
  {:ok, result} ->
    use_result(result)

  {:error, :safety_disarmed} ->
    alert_operator(:disarmed)

  {:error, {:command_failed, reason}} ->
    handle_failure(reason)
end
case BB.Jido.Action.Reactor.run(params, %{}) do
  {:ok, %{result: result}} ->
    use_result(result)

  {:error, {:reactor_halted, _halted}} ->
    handle_halt()

  {:error, {:reactor_failed, errors}} ->
    handle_failure(errors)
end
Why not structured BB.Error types?
BB itself uses Splode-backed error structs (BB.Error.*). bb_jido
deliberately uses tagged tuples instead because:
	Jido actions return {:ok, _} / {:error, _} tuples uniformly; that's
the expected idiom.
	Agent signal handlers pattern-match on the tag most of the time. A
tag is cheaper to match than a struct module.
	The underlying reason in {:command_failed, reason} is opaque — if
the command callback returned a structured error, it's still there
for inspection.

If your application standardises on BB.Error structs, wrap the action
boundary yourself.

